



Q & A

READERS' QUESTIONS, EDITORS' ANSWERS

Decoding GPS Output

Q This letter concerns Loran-C and GPS receivers and the NMEA 0183 data that they output. How can I interface them to a PC or Commodore computer? — G. C. B., Radford, VA

A If your computer needs to know where it is, try the interface circuit in Fig. 1 and the IBM PC BASIC program in Listing 1. NMEA-standard navigation receivers (both Loran and GPS) output a stream of serial data at 4800 baud for use by autopilots and other navigational equipment. You can feed that data stream directly to the serial port of a computer. We tried it with a Magellan 3000 XL GPS receiver.




FIG. 1—USE THIS DIAGRAM TO HELP you connect a GPS receiver (NMEA interface) to your PC. The orange wire is needed only for bi-directional communication outside the NMEA standard.

As soon as the receiver locks on a set of satellites, it begins to transmit a series of messages every couple of seconds. Each message is one line of data, beginning with "\$" and ending with return. The most useful message is the GLL message, shown in Fig. 2, which gives the latitude, longitude, and precise time of day.

When building the interface, check the pin numbers carefully because, of course, the plug is a mirror image of the socket. Note that pins 1, 4, 6, and 8 are tied together so the computer will start listening immediately, without waiting




FIG. 2—HERE'S THE STRUCTURE OF AN NMEA 0183C "GLL" (geographic latitude and longitude) message. NEMA 0183A is similar, but has fewer decimal places and omits the time of day.

LISTING 1

```
100 ' Program to decode NMEA 0183C data
110 ' M. Covington 1997
120 '
130 CLOSE #1 ' just in case
140 OPEN "COM1:4800,N,8,1" AS #1
150 PRINT "Press Ctrl-Break to stop"
160 ON ERROR GOTO 200 ' ignore comm errors
170 '
180 ' Get data from GPS receiver
190 PRINT
200 PRINT "Waiting for data..."
210 PRINT
220 LINE INPUT #1, D$
230 WHILE ASC(LEFT$(D$,1)) < 33 ' trim CR and LF
240 D$ = RIGHT$(D$,LEN(D$)-1)
250 WEND
260 PRINT
270 PRINT "Received: <";D$;">"
280 '
290 ' If it's a GLL message, decode it
300 IF LEFT$(D$,6) <> "$GPGLL" GOTO 190
310 LATD = VAL(MID$(D$,8,2)) ' lat degrees
320 LATM = VAL(MID$(D$,10,6)) ' lat minutes
330 LATX$ = MID$(D$,17,1) ' "N" or "S"
340 LOND = VAL(MID$(D$,19,3)) ' lon degrees
350 LONM = VAL(MID$(D$,22,6)) ' lon minutes
360 LONX$ = MID$(D$,29,1) ' "W" or "E"
370 UTCH = VAL(MID$(D$,31,2)) ' time hours
380 UTCM = VAL(MID$(D$,33,2)) ' time minutes
390 UTCS = VAL(MID$(D$,35,6)) ' time seconds
400 PRINT "Latitude ", LATD; " deg.", LATM; " min.", LATX$
410 PRINT "Longitude", LOND; " deg.", LONM; " min.", LONX$
420 PRINT "Time (UTC)", UTCH; ":", UTCM; ":", UTCS
430 GOTO 190
```

for any "ready" signal lines that are not used.

The NMEA data formats are described in some detail in the manual for

this Magellan GPS receiver, as well as on numerous Internet sites (do a search for "NMEA" on <http://www.altavista.com>). The full standard can be purchased for

\$125 from NMIEA, PO Box 3435, New
Bern, NC 28564.